package ch12
{
	import flash.display.Sprite;
	import flash.display.StageAlign;
	import flash.display.StageScaleMode;
	import flash.events.Event;
	import flash.geom.Point;
	import flash.utils.setTimeout;

	import utils.Ball;

	/**
	 * 万有引力
	 * @author sjp
	 *
	 */
	public class Gravity extends Sprite
	{
		private var particles:Array;
		private var numParticles:uint = 30;

		public function Gravity()
		{
			stage.scaleMode = StageScaleMode.NO_SCALE;
			stage.align = StageAlign.TOP_LEFT;
			setTimeout(init, 500);
		}

		private function init():void
		{

			particles = new Array();
			for (var i:uint = 0; i < numParticles; i++)
			{
				var size:Number = Math.random() * 25 + 5;
				var particle:Ball = new Ball(size);
				particle.x = Math.random() * stage.stageWidth;
				particle.y = Math.random() * stage.stageHeight;
				particle.mass = size;
				addChild(particle);
				particles.push(particle);
			}

			addEventListener(Event.ENTER_FRAME, onEnterFrame);
		}

		private function onEnterFrame(event:Event):void
		{
			for (var i:uint = 0; i < numParticles; i++)
			{
				var particle:Ball = particles[i];
				particle.x += particle.vx;
				particle.y += particle.vy;
			}
			for (i = 0; i < numParticles - 1; i++)
			{
				var partA:Ball = particles[i];
				for (var j:uint = i + 1; j < numParticles; j++)
				{
					var partB:Ball = particles[j];
					checkCollision(partA, partB);
					gravitate(partA, partB);
				}
			}
		}

		/**
		 * 引力
		 * @param partA
		 * @param partB
		 *
		 */
		private function gravitate(partA:Ball, partB:Ball):void
		{
			var dx:Number = partB.x - partA.x;
			var dy:Number = partB.y - partA.y;
			var distSQ:Number = dx * dx + dy * dy;
			var dist:Number = Math.sqrt(distSQ);
			/**
			 * force = G * m1 * m2 / (distance * distance);
			 * 简便计算:
			 * force = m1 * m2 / (distance * distance);
			 * */
			var force:Number = partA.mass * partB.mass / distSQ;
			/**
			 * 用 dx / dist 取代 Math.cos(angle)，
			 * 用 dy / dist 取代 Math.sin(angle)。
			 * 这样一来，就无需再使用 Math.atan2(dy, dx) 求出角度了。
			 *
			 * 原有方法，例子:
			 * var angle:Number=Math.atan2(dy, dx);
			 * var tx:Number=ball0.x + Math.cos(angle) * minDist;
			 * var ty:Number=ball0.y + Math.sin(angle) * minDist;
			 * */
			var ax:Number = force * dx / dist;
			var ay:Number = force * dy / dist;
			/**
			 * 根据物体的质量，需要把它们分解为两个部分。
			 * 想一想地球和太阳，两者之间存在着一种特殊的引力，是由质量除以距离的平方产生的。
			 * 地球与太阳在总引力的作用力下相互吸引。地球被太阳吸引，而太阳反过来又被地球吸引。
			 * 很明显，地球所获的加速度更多，因为它的质量比太阳小。所以，在系统中每个物体都有单独的加速度，用总加速度除以物体的质量。
			 * 因此，就有了最后四行公式。注意 partA 的加速度是增加的，而 partB 则是减少的。这要取决于 dx 和 dy 相减的顺序。
			 * */
			partA.vx += ax / partA.mass;
			partA.vy += ay / partA.mass;
			partB.vx -= ax / partB.mass;
			partB.vy -= ay / partB.mass;
		}

		/**
		 * 碰撞测试
		 * @param ball0
		 * @param ball1
		 *
		 */
		private function checkCollision(ball0:Ball, ball1:Ball):void
		{
			var dx:Number = ball1.x - ball0.x;
			var dy:Number = ball1.y - ball0.y;
			var dist:Number = Math.sqrt(dx * dx + dy * dy);
			if (dist < ball0.radius + ball1.radius)
			{
				// 计算角度和正余弦值 
				var angle:Number = Math.atan2(dy, dx);
				var sin:Number = Math.sin(angle);
				var cos:Number = Math.cos(angle);
				/**
				 * 旋转坐标至x轴水平，使得碰撞只与x轴有关，与y轴无关。
				 * 记住 vy 的值永远不变。
				 * */
				// 旋转 ball0 的位置 
				var pos0:Point = new Point(0, 0);
				// 旋转 ball1 的速度 
				var pos1:Point = rotate(dx, dy, sin, cos, true);
				// 旋转 ball0 的速度 
				var vel0:Point = rotate(ball0.vx, ball0.vy, sin, cos, true);
				// 旋转 ball1 的速度 
				var vel1:Point = rotate(ball1.vx, ball1.vy, sin, cos, true);
				// 碰撞的作用力
				var vxTotal:Number = vel0.x - vel1.x;
				vel0.x = ((ball0.mass - ball1.mass) * vel0.x + 2 * ball1.mass * vel1.x) / (ball0.mass + ball1.mass);
				vel1.x = vxTotal + vel0.x;

				/**解决两个物体黏在一起的问题.
				 * 首先确定绝对速度（自创的一个词），是所有速度的绝对值之和。
				 * 例如，如果一个速度是 -5 另一个速度是 10，那么绝对值就是 5 和 10，总和就是 5 + 10 = 15。
				 * 接下来，确定小球之间重叠部分的大小。用总半径长度减去总距离。
				 * 然后根据小球速度与绝对速度的百分比，让小球移动出重叠的那一部分。
				 *  结果会让小球之间没有重叠。这种方法比早前的版本要复杂一些，但确消除了很多 bug。
				 * */
				// 更新位置
				var absV:Number = Math.abs(vel0.x) + Math.abs(vel1.x);
				var overlap:Number = (ball0.radius + ball1.radius) - Math.abs(pos0.x - pos1.x);
				pos0.x += vel0.x / absV * overlap;
				pos1.x += vel1.x / absV * overlap;

				/**原有方法*/
				// 更新位置
				//				pos0.x += vel0.x;
				//				pos1.x += vel1.x;
				// 将位置旋转回来
				var pos0F:Object = rotate(pos0.x, pos0.y, sin, cos, false);
				var pos1F:Object = rotate(pos1.x, pos1.y, sin, cos, false);
				/**
				 * 最终的位置与系统中心点 ball0 的位置相对的。
				 * 因此，需要把它们都加到 ball0 的位置上，从而得到实际在屏幕上的位置。
				 * */
				// 将位置调整为屏幕的实际位置 
				ball1.x = ball0.x + pos1F.x;
				ball1.y = ball0.y + pos1F.y;
				ball0.x = ball0.x + pos0F.x;
				ball0.y = ball0.y + pos0F.y;
				// 将速度旋转回来 
				var vel0F:Object = rotate(vel0.x, vel0.y, sin, cos, false);
				var vel1F:Object = rotate(vel1.x, vel1.y, sin, cos, false);
				ball0.vx = vel0F.x;
				ball0.vy = vel0F.y;
				ball1.vx = vel1F.x;
				ball1.vy = vel1F.y;
			}
		}

		private function rotate(x:Number, y:Number, sin:Number, cos:Number, reverse:Boolean):Point
		{
			var result:Point = new Point();
			if (reverse)
			{
				result.x = x * cos + y * sin;
				result.y = y * cos - x * sin;
			}
			else
			{
				result.x = x * cos - y * sin;
				result.y = y * cos + x * sin;
			}
			return result;
		}
	}
}
